
* 
* LSTRUCT.PRG 
* 
* CA-Clipper 5.2 
* 
* 20.05.2001., 21.04.2008. 
* 
* formira STRUCT.TXT sa strukturom svih 
* *.DBF i *.DAT dBase (CA-Clipper) kompatibilnih datoteka 
* 
 
public arr[0],i,fil,fbr,dbs[0],db,j,drr,irr,fl 
 
fil:="STRUCT.TXT" 
arr:={} 
drr:=directory("*.DBF","HS") 
for i:=1 to len(drr) 
  aadd(arr,upper(alltrim(drr[i,1]))) 
next 
drr:=directory("*.DAT","HS") 
for i:=1 to len(drr) 
  aadd(arr,upper(alltrim(drr[i,1]))) 
next 
arr:=asort(arr) 
fbr:=fcreate(fil) 
for i:=1 to len(arr) 
   db:=ALLTRIM(UPPER(arr[i])) 
   use (db) exclusive new 
   fl:=left(arr[i],len(arr[i])-4) 
   irr:={} 
   drr:=directory(fl+"*.NTX","HS") 
   for j:=1 to len(drr) 
     aadd(irr,upper(alltrim(drr[j,1]))) 
   next 
   drr:=directory(fl+"*.IND","HS") 

   for j:=1 to len(drr) 
     aadd(irr,upper(alltrim(drr[j,1]))) 
   next 
   irr:=asort(irr) 
   for j:=1 to len(irr) 
     ordlistadd(irr[j]) 
   next 
   fwrite(fbr,CHR(13)+CHR(10)) 
   fwrite(fbr,"Database file: "+db+" ("+alltrim(str(lastrec(),8,0))+" recs)"+CHR(13)+CHR(10)) 
   fwrite(fbr,replicate("=",len("Database file: "+db+" ("+alltrim(str(lastrec(),8,0))+" recs)"))+CHR(13)+CHR(10)) 
   fwrite(fbr,CHR(13)+CHR(10)) 
   fwrite(fbr,"     Field Name  Type  Len  Dec"+CHR(13)+CHR(10)) 
   fwrite(fbr,"-------------------------------"+CHR(13)+CHR(10)) 
   dbs:=dbstruct() 
   for j:=1 to len(dbs) 
      fwrite(fbr,str(j,3,0)+"  "+padr(alltrim(upper(dbs[j,1])),14)+padr(upper(dbs[j,2]),4) ; 
            +str(dbs[j,3],3,0)+"  "+str(dbs[j,4],3,0)+CHR(13)+CHR(10)) 
   next 
   if len(irr)>0 
     fwrite(fbr,CHR(13)+CHR(10)) 
     fwrite(fbr,"    Index Name    Key"+CHR(13)+CHR(10)) 
     fwrite(fbr,"---------------------"+CHR(13)+CHR(10)) 
     for j:=1 to len(irr) 
       ordsetfocus(j) 
       fwrite(fbr,str(j,2,0)+"  "+padr(alltrim(upper(irr[j])),14)+upper(alltrim(indexkey()))+CHR(13)+CHR(10)) 
     next 
   endif 
   dbcloseall() 
   fwrite(fbr,CHR(13)+CHR(10)) 
next 
quit 


